Cross-Over Trials with Covariance Pattern Models

As mentioned at the beginning of the cross-over section, cross-over experiments are a special case of a repeated measures experiment. Up to this point, we have assumed that the observations within a patient have the same correlation and variance, ignoring the possibility that observations from periods closer together may be more correlated than observations from periods further apart or that the treatments may have heterogeneous variances. Therefore, there are two possible ways to structure the covariance patterns, by period and by treatment. We will now go through an example of fitting a cross-over study with covariance patterns that is presented in our text.

Four-way cross-over trial fitting covariance patterns

This trial is one originally described by Jones and Kenward (1989). There are four treatments, A, C, D and placebo B. Cardiac output, measured by the left ventricular ejection time was compared among the four treatments. There were four periods, each a week long, with a week-long washout period between the treatment periods. Cardiac output was measured at the end of each treatment period. There are data on 14 patients (56 observations). Except for two patients, each patient represents a sequence, so a sequence effect is not included in the model. The linear model for this experiment

$$y_{ij} = \mu + \xi_i + \pi_j + \tau_{t(j)} + \delta_j \lambda_{r(j)} + e_{ij}$$

μ = mean
ξ_i = random subject effect
π_j = period effect for period j
$\tau_{t(j)}$= direct treatment effect for treatment t
$\lambda_{r(j)}$ = the first-order "carryover" effect due to treatment r, where r is the treatment applied the previous period
δ_j = an indicator variable: 0 if it is the first period
1 if the period is greater than 1

$\underset{\sim}{e}_i \square N(0, R)$ where R is 4x4 covariance matrix for each subject, where each variance refers to a period when the analysis is structured by period and refers to a treatment when the analysis is structured by treatment.

We will first run the analysis without fitting a covariance pattern. The SAS code for that analysis follows.

```
*run the analysis without fitting a covariance pattern;
PROC MIXED noclprint;
CLASS treata period patient carry;
MODEL lvet=treata period carry*delta/ CHISQ DDFM=satterth;
Random patient;
estimate 'carry A-B' carry*delta 1 -1 0 0;
estimate 'carry A-C' carry*delta 1 0 -1 0;
estimate 'carry A-D' carry*delta 1 0 0 -1;
estimate 'carry B-C' carry*delta 0 1 -1 0;
```

```
estimate 'carry B-D' carry*delta 0 1 0 -1;
estimate 'carry C-D' carry*delta 0 0 1 -1;
LSMEANS treata/DIFF PDIFF;
run;
```

Following are selected output from this analysis.

Covariance Parameter Estimates

Cov Parm	Estimate
patient	1676.83
Residual	1886.92

Fit Statistics

-2 Res Log Likelihood	519.2
AIC (smaller is better)	523.2
AICC (smaller is better)	523.5
BIC (smaller is better)	524.5

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
treata	3	33.2	17.42	5.81	0.0006	0.0026
period	2	33.1	0.10	0.05	0.9511	0.9512
delta*carry	3	34.6	2.21	0.74	0.5309	0.5383

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
carry A-B	-27.4544	20.4861	34.2	-1.34	0.1890
carry A-C	-23.3652	21.0909	34.5	-1.11	0.2756
carry A-D	-8.8551	20.9692	34.5	-0.42	0.6754
carry B-C	4.0892	21.5281	34.6	0.19	0.8505
carry B-D	18.5993	22.3821	34.7	0.83	0.4117
carry C-D	14.5101	22.3836	34.9	0.65	0.5211

Least Squares Means

Effect	treata	Estimate	Standard Error	DF	t Value	Pr > |t|
treata	A	384.55	16.2638	31.8	23.64	<.0001
treata	B	346.54	16.4366	32.5	21.08	<.0001
treata	C	371.14	16.7128	33.7	22.21	<.0001
treata	D	315.87	16.8019	34.1	18.80	<.0001

Differences of Least Squares Means

Effect	treata	_treata	Estimate	Standard Error	DF	t Value	Pr > |t|
treata	A	B	38.0041	17.3831	33.1	2.19	0.0360
treata	A	C	13.4065	17.8700	33.2	0.75	0.4584
treata	A	D	68.6757	17.5174	33.2	3.92	0.0004
treata	B	C	-24.5976	17.8630	33.3	-1.38	0.1777
treata	B	D	30.6716	18.2393	33.3	1.68	0.1020

treata	C	D	55.2692	18.6644	33.4	2.96	0.0056

We can see from this analysis, that there are significant differences in the effects of treatment on the measure of cardiac output. Neither period nor carryover effects appear to differ significantly. However, we are not able to say anything about possible differences in correlations among the periods, differences in variation amongst the periods or differences in variation amongst the treatments. The first model that we will look at will be to look at the covariance pattern structure applied to periods. The following SAS program fits that model assuming the unstructured or general covariance pattern.

```
title 'Unstructured or General';
PROC MIXED noclprint;
CLASS treata period patient carry;
MODEL lvet=treata period carry*delta/ CHISQ DDFM=satterth;
REPEATED period/ SUBJECT=patient TYPE=un rcorr;
estimate 'carry A-B' carry*delta 1 -1 0 0;
estimate 'carry A-C' carry*delta 1 0 -1 0;
estimate 'carry A-D' carry*delta 1 0 0 -1;
estimate 'carry B-C' carry*delta 0 1 -1 0;
estimate 'carry B-D' carry*delta 0 1 0 -1;
estimate 'carry C-D' carry*delta 0 0 1 -1;
LSMEANS treata/DIFF PDIFF;
ods output covparms=cov;
ods output rcorr=corr;
run;
```

Just as we have done in previous repeated measures study, we want to investigate the covariance and correlations to determine what covariance pattern might fit the best. The following SAS code creates the datasets and the plots.

```
data times;
do time1=1 to 4;
 do time2=1 to time1;
  dist=time1-time2;
  output;
  end;
  end;
  run;

  data covplot;
  merge times cov;
  proc print;
run;

axis1 value=(font=swiss2 h=2) label=(angle=90 f=swiss h=2 'Covariance of
between Subj effects');
axis2 value=(font=swiss h=2) label=(f=swiss h=2 'Distance');
legend1 value=(font=swiss h=2) label=(f=swiss h=2 'From Period');
symbol1 color=black interpol=join line=1 value=square;
symbol2 color=black interpol=join line=2 value=circle;
symbol3 color=black interpol=join line=20 value=triangle;
symbol4 color=black interpol=join line=4 value=star;
proc gplot data=covplot;
plot estimate*dist=time2/vaxis=axis1 haxis=axis2 legend=legend1;
```

```
run;
data corrlplot;
 set corr;
  if row=1 then do;
   dist=1; corr=col2; output;
   dist=2; corr=col3; output;
   dist=3; corr=col4; output;
  end;
  if row=2 then do;
   dist=1; corr=col3; output;
   dist=2; corr=col4; output;
  end;
  if row=3 then do;
   dist=1; corr=col4; output;
  end;
proc gplot data=corrlplot;
 plot corr*dist=row/haxis=axis1 vaxis=axis3 legend=legend1;;
run;
```



From these figures we note that there appears to be differences in the variances among the periods. Although there may be a possibility that the further apart the periods, the smaller the correlation, the small range in the correlations may suggest compound symmetry. So possible covariance structures to investigate and compare with the unstructured would be compound symmetry and heterogeneous Toeplitz. The following table compiles the results from the unstructured, compound symmetry, and heterogeneous Toeplitz models under the period structure.

Covariance Pattern	**-2 log likelihood**	**parameters**	**AIC**	**AICC**	**BIC**
Unstructured	502.2	10	522.2	528.5	528.6
Compound Symmetry	519.2	2	523.2	523.5	524.5
Heterogeneous Toeplitz	503.1	7	517.1	520.0	521.6

It appears that the heterogeneous Toeplitz covariance structure fits the best when looking across periods. We will now look across treatments. Certain covariance patterns would not

make sense when looking across treatment. The AR(1) and Toeplitz covariance patterns are used when the distance in time between measures changes. With treatments, we are not comparing distances, but treatments. Therefore, the unstructured covariance pattern and possibly compound symmetry would be reasonable patterns. We have the results for the compound symmetry in the table above. These don't change even though we are looking The SAS code is exactly the same as for period, except that the repeated statement now has "repeated treat" in place of "repeated period". The following table summarizes the analysis.

Covariance Pattern	-2 log likelihood	parameters	AIC	AICC	BIC
Unstructured	498.0	10	518.0	524.3	524.4

Because the compound symmetry covariance model is nested within both the unstructured models, we can test for significant differences in fit with the likelihood ratio test.

Period – Unstructured vs. CS

519.2-502.2=17.0, for χ^2 with 8 df, p ~ .04

Treatment – Unstructured vs. CS

519.2-498.0=21.8, for χ^2 with 8 df, p~ .007

Both unstructured covariance pattern models show significant improvement over the compound symmetry covariance pattern model.

We can compare the variances and correlations of the unstructured covariance patterns for the across period and across treatment analyses.

Period	Variance	Correlations			
1	6898.45	1			
2	1931.71	0.6748	1		
3	3930.36	0.4056	0.6784	1	
4	1186.20	0.5262	0.4811	0.5244	1
Treatment					
A	1681.02	1			
B	6360.25	0.09861	1		
C	3429.52	0.1628	0.8333	1	
D	2531.98	-0.09949	0.6367	0.8184	1

The first thing to note in the period structure is that the variance in the first period is much greater than in the subsequent periods. This is a common occurrence and is why an adjustment period is sometimes used. For the treatment structure model, Treatment B, which is the placebo, has greater variation than any of the three active treatments. Also note that for treatment A, the correlations with the other treatments are much lower than between the other three treatments.

Finally, we can compare the estimated treatment differences and carryover differences from both models. First the SAS output for the tests of fixed effects from the two models.

Period structure:

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
treata	3	21.1	29.12	9.71	<.0001	0.0003
period	2	10.7	0.18	0.09	0.9144	0.9151
delta*carry	3	18.3	6.27	2.09	0.0991	0.1367

Treatment structure:

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
treata	3	11.5	37.81	12.60	<.0001	0.0006
period	2	14.8	0.77	0.39	0.6790	0.6857
delta*carry	3	16.4	11.19	3.73	0.0107	0.0324

With both structures there are significant treatment differences. However, the carryover effect is only significant with the treatment structure and not the period structure. Part of this difference may be due to the small number of patients in the study. If there were more patients, then the two structures would be in better agreement. Following are the estimated treatment differences and carryover differences.

Period structure:

Differences of Least Squares Means

Effect	treata	_treata	Estimate	Standard Error	DF	t Value	Pr > |t|
treata	A	B	37.1182	13.2412	22.5	2.80	0.0102
treata	A	C	23.6274	14.4131	23.1	1.64	0.1147
treata	A	D	64.0797	12.5656	21.4	5.10	<.0001
treata	B	C	-13.4907	13.1388	19.8	-1.03	0.3169
treata	B	D	26.9615	14.0343	22.6	1.92	0.0674
treata	C	D	40.4523	13.5273	21.3	2.99	0.0069

Treatment structure:

Differences of Least Squares Means

Effect	treata	_treata	Estimate	Standard Error	DF	t Value	Pr > |t|
treata	A	B	43.8265	23.2397	12.4	1.89	0.0829
treata	A	C	8.3143	18.3950	12.1	0.45	0.6593
treata	A	D	66.3686	18.4938	9.8	3.59	0.0051
treata	B	C	-35.5123	13.3787	13.3	-2.65	0.0195
treata	B	D	22.5421	17.2364	14.1	1.31	0.2119
treata	C	D	58.0543	11.0837	15.3	5.24	<.0001

Period structure:

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
carry A-B	-20.0486	13.9013	18.3	-1.44	0.1661

carry A-C	-35.0836	14.7364	17.5	-2.38	0.0289
carry A-D	-5.6223	13.8226	16.7	-0.41	0.6894
carry B-C	-15.0350	14.2315	18.8	-1.06	0.3041
carry B-D	14.4262	15.5769	19	0.93	0.3660
carry C-D	29.4612	15.3691	19.8	1.92	0.0698

Treatment structure:

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|
carry A-B	-23.5450	12.1550	17	-1.94	0.0695
carry A-C	-54.0162	16.4769	17.6	-3.28	0.0043
carry A-D	-4.2811	11.1641	7.98	-0.38	0.7114
carry B-C	-30.4712	15.5719	17	-1.96	0.0670
carry B-D	19.2639	14.1348	11.6	1.36	0.1987
carry C-D	49.7351	18.0074	30.2	2.76	0.0097

First, comparing the differences in the treatments between the period structure and treatment structure, we notice differences in the levels of significance among the treatment comparisons. For example, the BC comparison in the period structure shows the difference (-13.5) is not statistically significant, while the BC comparison in the period structure shows the difference (-35.5) to be statistically significant. The other comparisons that involve Treatment C also vary more than the other comparisons. We see this variation with Treatment C comparisons in the carryover effect, also. Which of the two models should be used? The treatment structured model appears to fit the best as seen by the smaller -2 log likelihood and the greater level of significance. However, we see that whether or not the differential carryover effect accounts for a significant amount of the variability in the response is dependent on the covariance structure chosen. Because this trial was designed to try and avoid possible carryover with washout periods between treatments, the probability that the carryover effect is real is questionable. The authors felt that it was advisable to drop the carryover effect from the model and rerun the analyses. We will present these analysis in the next section.